```
AppendTo[$Path, ToFileName[{$HomeDirectory, "Dropbox", "mathematica"}]];
<< TrigInt`
```

Upper bounds

```
T2[a_, b_] = {{-1, 0}, {1, 0}, {a, b}}; (* general triangle *)
f = Equilateral[Neumann, Symmetric][0, 1]
f = Transplant[f, {{0, Sqrt[3]}, {1, 0}, {-1, 0}}, Equilateral[]]
(* scale and rotate to put nodal line near vertex (-1,0) *)
-(D[ef1, x, x] + D[ef1, y, y]) / ef1 // FullSimplify
And @@ (#[[2]] ≤ #[[1]] ≤ #[[3]] & /@ {Limits[T2[0, Sqrt[3]]]})
RegionPlot[{f > 0, %}, {x, -2, 2}, {y, -2, 2}]
tf = Transplant[f, T2[a, b], T2[0, Sqrt[3]]] // FullSimplify
TrigInt[tf, Limits[T2[a, b]]] // FullSimplify[#, b > 0] &
(* average is zero, good test function *)
U1 = Rayleigh[tf, T2[a, b]] // FullSimplify[#, b > 0] &
(* U1 is the upper bound for near equilateral *)
U2 = BesselJZero[0, 1]^2
(* U2 is the upper bound for near degenerate (Cheng's bound)
 We could also use Rayleigh[x-a/3,T2[a,b]]. *)
lin = x; (*+b/2 y;*)
c = TrigInt[lin, Limits[T2[a, b]]] / b // FullSimplify[#, b > 0] &
TrigInt[lin - c, Limits[T2[a, b]]] // FullSimplify[#, b > 0] &
U2 = Rayleigh[lin - c, T2[a, b]] // FullSimplify[#, b > 0] &
```

$$-\cos\left(\frac{1}{3}\pi(2x+2)\right)\cos\left(\frac{\pi\left(\sqrt{3}-2y\right)}{\sqrt{3}}\right)+\cos\left(\frac{1}{3}\pi(2x-1)\right)\cos\left(\pi\left(1-\frac{2y}{\sqrt{3}}\right)\right)-\cos\left(\frac{1}{3}\pi(5-4x)\right)$$

$$-\cos\left(\frac{1}{3}\pi\left(5-4\left(\frac{x}{4}-\frac{\sqrt{3}\,y}{4}+\frac{3}{4}\right)\right)\right)+\cos\left(\pi\left(1-\frac{2\left(-\frac{\sqrt{3}\,x}{4}-\frac{y}{4}+\frac{\sqrt{3}}{4}\right)}{\sqrt{3}}\right)\right)\cos\left(\frac{1}{3}\pi\left(2\left(\frac{x}{4}-\frac{\sqrt{3}\,y}{4}+\frac{3}{4}\right)-1\right)\right)-$$

$$\cos\left(\frac{\pi\left(\sqrt{3}-2\left(-\frac{\sqrt{3}\,x}{4}-\frac{y}{4}+\frac{\sqrt{3}}{4}\right)\right)}{\sqrt{3}}\right)\cos\left(\frac{1}{3}\pi\left(2\left(\frac{x}{4}-\frac{\sqrt{3}\,y}{4}+\frac{3}{4}\right)+2\right)\right)$$

0

$$0\le y\le\sqrt{3}\bigwedge\frac{y}{\sqrt{3}}-1\le x\le 1-\frac{y}{\sqrt{3}}$$



$$\sin\left(\frac{\pi\,(2\,(a+3)\,y-2\,b\,x+b)}{6\,b}\right)-2\sin\left(\frac{\pi\,(-a\,y+b\,x+y)}{2\,b}\right)\cos\left(\frac{\pi\,(b\,(x+1)-(a+3)\,y)}{6\,b}\right)$$

$$0$$

$$\frac{64\,\pi^2\,(a^2+b^2+3)+243\,((a-6)\,a+b^2-3)}{288\,b^2}$$

$$(j_{0,1})^2$$

$$\frac{a}{3}$$

$$0$$

$$\frac{18}{a^2+3}$$

Lower bounds

```
(* (0,Sqrt[3]) as reference *)
ineq1 = (a^2 + 3) (1 - γ) + 2 a b δ + b^2 γ < 4 π^2 / 3 / u;
(* (0,1) as reference *)
ineq2 = (a^2 + 1) (1 - γ) + 2 a b δ + b^2 γ < 3 π^2 / 4 / u;
(* (1/2,Sqrt[3]/2) as reference (half of equilateral) *)
ineq3 = ((a - 1 / 2) ^2 + 3 / 4) (1 - γ) + 2 (a - 1 / 2) b δ + b^2 γ < 2 π^2 / 3 / u;
(* (1,2/Sqrt[3]) as reference (half of equilateral, longest side on x-axis )*)
ineq4 = ((a - 1) ^2 + 4 / 3) (1 - γ) + 2 (a - 1) b δ + b^2 γ < 8 π^2 / 9 / u;
(* Near equilateral we need 1 2 3. Near degenerate 3 4. *)
neareq = ineq1 || ineq2 || ineq3 /. u → U1
nearde = ineq3 || ineq4 /. u → U2
middle = ineq3 || ineq4 /. u → U1
g1[c_?NumberQ, d_?NumberQ] :=
 FullSimplify[neareq /. a → c /. b → d, 0 ≤ γ ≤ 1 && -1 / 2 ≤ δ ≤ 1 / 2]
g2[c_?NumberQ, d_?NumberQ] :=
 FullSimplify[nearde /. a → c /. b → d, 0 ≤ γ ≤ 1 && -1 / 2 ≤ δ ≤ 1 / 2]
```

$$\left(a^2+3\right)(1-\gamma)+2\,a\,b\,\delta+b^2\,\gamma<\frac{384\,\pi^2\,b^2}{64\,\pi^2\,(a^2+b^2+3)+243\,((a-6)\,a+b^2-3)}\,\bigvee$$

$$\left(a^2+1\right)(1-\gamma)+2\,a\,b\,\delta+b^2\,\gamma<\frac{216\,\pi^2\,b^2}{64\,\pi^2\,(a^2+b^2+3)+243\,((a-6)\,a+b^2-3)}\,\bigvee$$

$$2\left(a-\frac{1}{2}\right)b\,\delta+\left(\left(a-\frac{1}{2}\right)^2+\frac{3}{4}\right)(1-\gamma)+b^2\,\gamma<\frac{192\,\pi^2\,b^2}{64\,\pi^2\,(a^2+b^2+3)+243\,((a-6)\,a+b^2-3)}$$

$$2\left(a-\frac{1}{2}\right)b\,\delta+\left(\left(a-\frac{1}{2}\right)^2+\frac{3}{4}\right)(1-\gamma)+b^2\,\gamma<\frac{1}{27}\,\pi^2\left(a^2+3\right)\bigvee 2\,(a-1)\,b\,\delta+\left((a-1)^2+\frac{4}{3}\right)(1-\gamma)+b^2\,\gamma<\frac{4}{81}\,\pi^2\left(a^2+3\right)$$

$$2\left(a-\frac{1}{2}\right)b\,\delta+\left(\left(a-\frac{1}{2}\right)^2+\frac{3}{4}\right)(1-\gamma)+b^2\,\gamma<\frac{192\,\pi^2\,b^2}{64\,\pi^2\,(a^2+b^2+3)+243\,((a-6)\,a+b^2-3)}\,\bigvee$$

$$2\,(a-1)\,b\,\delta+\left((a-1)^2+\frac{4}{3}\right)(1-\gamma)+b^2\,\gamma<\frac{256\,\pi^2\,b^2}{64\,\pi^2\,(a^2+b^2+3)+243\,((a-6)\,a+b^2-3)}$$

```
tri = (a + 1) ^2 + b^2 ≤ 4 && a > 0 && 2 a^2 + 2 b^2 > 2;
RegionPlot[{g1[a, b], g2[a, b], tri, a > 1 / 2},
 {a, 0, 1}, {b, 0, Sqrt[3]}, PlotPoints → 10, MaxRecursion → 2]
```

$(a+1)^2 + b^2 \leq 4 \wedge a > 0 \wedge 2\,a^2 + 2\,b^2 > 2$



Clearly neareq and nearde cover every acute triangle.

Proof for nearly degenerate triangles (with 1/2 ≤ a ≤ 1 and 0 < b ≤ 1.08).

```
RegionPlot[{tri, 1 / 2 ≤ a ≤ 1 && 0 < b ≤ 1.08}, {a, 0, 1}, {b, 0, Sqrt[3]}]
```




```
nearde (* both inequalities *)
```

$$2\left(a-\frac{1}{2}\right) b\,\delta + \left(\left(a-\frac{1}{2}\right)^2 + \frac{3}{4}\right)(1-\gamma) + b^2\,\gamma < \frac{1}{27}\,\pi^2\left(a^2+3\right) \bigvee 2\,(a-1)\,b\,\delta + \left((a-1)^2 + \frac{4}{3}\right)(1-\gamma) + b^2\,\gamma < \frac{4}{81}\,\pi^2\left(a^2+3\right)$$

At least one inequality will be true if a positive linear combination of inequalities is true.

```
nearde /. Less → Subtract /. Or → List
left = {1 - a, a - 1 / 2}.% // Collect[#, γ, FullSimplify] &
  (* linear combination, δ eliminated, linear in γ *)
```

$$\left\{-\frac{1}{27}\pi^2\left(a^2+3\right)+2\left(a-\frac{1}{2}\right)b\,\delta+\left(\left(a-\frac{1}{2}\right)^2+\frac{3}{4}\right)(1-\gamma)+b^2\gamma,\ -\frac{4}{81}\pi^2\left(a^2+3\right)+2(a-1)\,b\,\delta+\left((a-1)^2+\frac{4}{3}\right)(1-\gamma)+b^2\gamma\right\}$$

$$-\frac{1}{81}\pi^2(a+1)\left(a^2+3\right)+\frac{1}{6}\gamma\left(a(3a-8)+3b^2+1\right)+\frac{1}{6}((8-3a)a-1)$$

Linear in γ, hence we need to check endpoints $\gamma = 0$ and $\gamma = 1$.

```
both = {left < 0 /. γ → 1, left < 0 /. γ → 0} // Collect[#, {a, b}, FullSimplify] &
```

$$\left\{-\frac{1}{81}\pi^2 a^3-\frac{\pi^2 a^2}{81}-\frac{\pi^2 a}{27}+\frac{b^2}{2}-\frac{\pi^2}{27}<0,\ -\frac{1}{81}\pi^2 a^3+\left(-\frac{1}{2}-\frac{\pi^2}{81}\right)a^2-\frac{1}{27}(\pi-6)(6+\pi)\,a+\frac{1}{54}\left(-9-2\pi^2\right)<0\right\}$$

The first inequality has negative signs for a, so we can put a = 1/2. The second inequality is a cubic in a.

```
both[[1]] /. a → 1 / 2 // Reduce // N
both[[2]] // Reduce // N
```

$-1.08996 < b < 1.08996$

$a > -6.44176$

Hence we are done.

Algorithm for polynomial inequalities

PolyNeg[P,{x,y},{dx,dy}]: show that polynomial P with variables x, y is nonpositive on rectangle (0, 0) (dx, dy). True means it is, False means algorithm failed. See http://www.ams.org/mathscinet-getitem?mr=MR2779073 for detail.

We can use this on large polynomials in all other cases.

```
CumFun[f_, l_] := Rest[FoldList[f, 0, l]];
PolyNeg[P_, {x_, y_}, {dx_, dy_}] :=
  ((Fold[CumFun[Min[#1, 0] / dy + #2 &, Map[Max[#1, 0] &, #1] dx + #2] &,
      0, Reverse[CoefficientList[P, {x, y}]]] // Max) ≤ 0);
```

Proof for triangles with 1/2 ≤ a ≤ 1 and b ≥ 1.08.

```
neareq
```

$$\left(a^2+3\right)(1-\gamma)+2\,a\,b\,\delta+b^2\gamma<\frac{384\,\pi^2 b^2}{64\,\pi^2\left(a^2+b^2+3\right)+243\left((a-6)\,a+b^2-3\right)}\ \bigvee$$
$$\left(a^2+1\right)(1-\gamma)+2\,a\,b\,\delta+b^2\gamma<\frac{216\,\pi^2 b^2}{64\,\pi^2\left(a^2+b^2+3\right)+243\left((a-6)\,a+b^2-3\right)}\ \bigvee$$
$$2\left(a-\frac{1}{2}\right)b\,\delta+\left(\left(a-\frac{1}{2}\right)^2+\frac{3}{4}\right)(1-\gamma)+b^2\gamma<\frac{192\,\pi^2 b^2}{64\,\pi^2\left(a^2+b^2+3\right)+243\left((a-6)\,a+b^2-3\right)}$$

Coefficient for δ is positive, hece we can put $\delta = 1/2$. Now we need to show that at least one inequolity is true for any γ. Start with inequality 3.

```
neareq /. Or → List /. Less → Subtract;
lin = %[[3]] /. δ → 1 / 2 // Collect[#, γ, FullSimplify] &
RegionPlot[{D[lin, γ] > 0, tri && b > 1.08 && a > 1 / 2}, {a, 0, 1}, {b, 0, Sqrt[3]}]
(* red are is where we need a proof, blue has positive coefficient for γ *)
D[lin, γ]
```

$$\gamma\left(-a^2 + a + b^2 - 1\right) - \frac{192\,\pi^2\, b^2}{64\,\pi^2\,(a^2 + b^2 + 3) + 243\,((a - 6)\, a + b^2 - 3)} + a^2 + a\,(b - 1) - \frac{b}{2} + 1$$



$$-a^2 + a + b^2 - 1$$

Derivative is clearly minimal for b = 1.08, and it has no zeros in the red area. Hence we need to check only large values of γ. Unfortunately γ=1 does not work.

```
lin /. γ → 2 / 3
RegionPlot[Evaluate[{% < 0, tri && b > 1.08 && a > 1 / 2}], {a, 0, 1}, {b, 0, Sqrt[3]}]
```

$$\frac{2}{3}\left(-a^2+a+b^2-1\right)-\frac{192\,\pi^2\,b^2}{64\,\pi^2\,(a^2+b^2+3)+243\,((a-6)\,a+b^2-3)}+a^2+a\,(b-1)-\frac{b}{2}+1$$



It seems that inequality is true. We will use PolyNeg to prove this. First we move the origin to the vertex with a=1/2.

```
Together[lin /. γ → 2 / 3 /. b → Sqrt[7] / 2 - b /. a → a + 1 / 2]
poly = poly2 = Numerator[%] Denominator[%];
PolyNeg[poly, {b, a}, {1 / 4, 99 / 1000}]
poly = poly /. a → a + 99 / 1000 /. b → b + 1 / 9;
PolyNeg[poly, {b, a}, {15 / 100, 1 / 10}]
```

$$\Big(256\,\pi^2\,a^4+972\,a^4-2916\,a^3\,b-768\,\pi^2\,a^3\,b+384\sqrt{7}\,\pi^2\,a^3+256\,\pi^2\,a^3+1458\sqrt{7}\,a^3-4860\,a^3+2916\,a^2\,b^2+768\,\pi^2\,a^2\,b^2-2916\sqrt{7}\,a^2\,b+14\,580\,a^2\,b-768\sqrt{7}\,\pi^2\,a^2\,b-768\,\pi^2\,a^2\,b+384\sqrt{7}\,\pi^2\,a^2+2368\,\pi^2\,a^2-7290\sqrt{7}\,a^2+243\,a^2-2916\,a\,b^3-768\,\pi^2\,a\,b^3+4374\sqrt{7}\,a\,b^2-9720\,a\,b^2+1152\sqrt{7}\,\pi^2\,a\,b^2+512\,\pi^2\,a\,b^2+9720\sqrt{7}\,a\,b+1458\,a\,b-512\sqrt{7}\,\pi^2\,a\,b-6528\,\pi^2\,a\,b+1920\sqrt{7}\,\pi^2\,a+1088\,\pi^2\,a-5832\sqrt{7}\,a-20\,655\,a+1944\,b^4+512\,\pi^2\,b^4-3888\sqrt{7}\,b^3-1024\sqrt{7}\,\pi^2\,b^3+9963\,b^2+4928\,\pi^2\,b^2+3645\sqrt{7}\,b-1344\sqrt{7}\,\pi^2\,b+1408\,\pi^2-16\,524\Big)\Big/\left(12\left(64\,\pi^2\,a^2+243\,a^2+64\,\pi^2\,a-1215\,a+243\,b^2+64\,\pi^2\,b^2-243\sqrt{7}\,b-64\sqrt{7}\,\pi^2\,b+320\,\pi^2-972\right)\right)$$

True

True

Hence inequality is true on the whole parameter space. Below is the plot of the rectangles used for the algorithm.

```
params = tri && b > 1.08 && a > 1 / 2 /. b → Sqrt[7] / 2 - b /. a → a + 1 / 2;
(* transformed parameter space *)
RegionPlot[{poly2 < 0, params, 0 < a < 99 / 1000 && 0 < b < 1 / 4,
  99 / 1000 < a < 199 / 1000 && 1 / 9 < b < 1 / 9 + 15 / 100}, {a, -0.1, 0.25}, {b, -0.05, 0.3}]
```



Now we need to handle 1 >= γ > 2/3. This time we will take a linear combination of inequalities 1 and 3.

```
neareq /. Or → List /. Less → Subtract
lin = %[[3]] + (b - a) %[[1]] /. δ → 1 / 2 // Collect[#, γ, FullSimplify] &
RegionPlot[{D[%, γ] > 0, tri && b > 1.08 && a > 1 / 2}, {a, 0, 1}, {b, 0, Sqrt[3]}]
D[lin, γ]
```

$$\left\{-\frac{384\,\pi^2\,b^2}{64\,\pi^2\,(a^2+b^2+3)+243\,((a-6)\,a+b^2-3)}+\left(a^2+3\right)(1-\gamma)+2\,a\,b\,\delta+b^2\,\gamma,\right.$$
$$-\frac{216\,\pi^2\,b^2}{64\,\pi^2\,(a^2+b^2+3)+243\,((a-6)\,a+b^2-3)}+\left(a^2+1\right)(1-\gamma)+2\,a\,b\,\delta+b^2\,\gamma,$$
$$\left.-\frac{192\,\pi^2\,b^2}{64\,\pi^2\,(a^2+b^2+3)+243\,((a-6)\,a+b^2-3)}+2\left(a-\frac{1}{2}\right)b\,\delta+\left(\left(a-\frac{1}{2}\right)^2+\frac{3}{4}\right)(1-\gamma)+b^2\,\gamma\right\}$$

$$\left(-64\,\pi^2\left(-a\left(a^2+3\right)b+a\,((a-1)\,a+4)\left(a^2+3\right)-a\,b^4-(a-6)\,b^3-(a\,(a+5)-3)\,b^2\right)-243\,a\left((a-6)\,a+b^2-3\right)((a-1)\,a-b\,(b+1)+4)\right)\Big/\left(64\,\pi^2\left(a^2+b^2+3\right)+243\left((a-6)\,a+b^2-3\right)\right)+\gamma\left(a^3-a^2\,(b+1)-a\left(b^2-4\right)+b\left(b^2+b-3\right)-1\right)+\frac{5\,b}{2}+1$$



$$a^3-a^2\,(b+1)-a\left(b^2-4\right)+b\left(b^2+b-3\right)-1$$

Here it is not so clear that coefficient for γ is negative, but we can use the algorithm to prove this. Then we can put γ=2/3.

```
lin /. γ → 2 / 3
RegionPlot[{% < 0, tri && b > 1.08 && a > 1 / 2}, {a, 0, 1}, {b, 0, Sqrt[3]}]
```

$$\left(-64\,\pi^2\left(-a\left(a^2+3\right)b+a\left((a-1)\,a+4\right)\left(a^2+3\right)-a\,b^4-(a-6)\,b^3-\left(a\,(a+5)-3\right)b^2\right)-243\,a\left((a-6)\,a+b^2-3\right)\left((a-1)\,a-b\,(b+1)+4\right)\right)\Big/\left(64\,\pi^2\left(a^2+b^2+3\right)+243\left((a-6)\,a+b^2-3\right)\right)+\frac{2}{3}\left(a^3-a^2\,(b+1)-a\left(b^2-4\right)+b\left(b^2+b-3\right)-1\right)+\frac{5\,b}{2}+1$$



Hence it appears that inequality is true.

```
poly = D[lin, γ] /. b → Sqrt[7] / 2 - b /. a → a + 1 / 2
PolyNeg[poly, {b, a}, {1 / 4, 1 / 5}]
```

$$-\left(a+\frac{1}{2}\right)^2\left(-b+\frac{\sqrt{7}}{2}+1\right)-\left(a+\frac{1}{2}\right)\left(\left(\frac{\sqrt{7}}{2}-b\right)^2-4\right)+\left(a+\frac{1}{2}\right)^3+\left(\frac{\sqrt{7}}{2}-b\right)\left(\left(\frac{\sqrt{7}}{2}-b\right)^2-b+\frac{\sqrt{7}}{2}-3\right)-1$$

True

Hence derivative is negative.

```
params = tri && b > 1.08 && a > 1 / 2 /. b → Sqrt[7] / 2 - b /. a → a + 1 / 2;
(* transformed parameter space *)
RegionPlot[{poly < 0, params, 0 < a < 1 / 5 && 0 < b < 1 / 4}, {a, -0.1, 0.25}, {b, -0.05, 0.3}]
```



```
Together[lin /. γ → 2 / 3 /. b → Sqrt[7] / 2 - b /. a → a + 1 / 2]
poly = poly2 = Numerator[%] Denominator[%];
PolyNeg[poly, {b, a}, {1 / 4, 5 / 100}]
poly = poly /. a → a + 5 / 100 /. b → b + 5 / 100;
PolyNeg[poly, {b, a}, {1 / 4 - 5 / 100, 9 / 100}]
poly = poly /. a → a + 9 / 100 /. b -> b + 12 / 100;
PolyNeg[poly, {b, a}, {1 / 4 - 17 / 100, 6 / 100}]
```

$$\Big(-128\pi^2 a^5 - 486 a^5 + 972 a^4 b + 256\pi^2 a^4 b - 128\sqrt{7}\,\pi^2 a^4 - 192\pi^2 a^4 - 486\sqrt{7}\,a^4 + 2187 a^4 - 5346 a^3 b + 128\pi^2 a^3 b - 64\sqrt{7}\,\pi^2 a^3 - 960\pi^2 a^3 + 2673\sqrt{7}\,a^3 + 2187 a^3 - 1458 a^2 b^2 + 384\pi^2 a^2 b^2 + 1458\sqrt{7}\,a^2 b - 4374 a^2 b - 384\sqrt{7}\,\pi^2 a^2 b + 384\pi^2 a^2 b - 192\sqrt{7}\,\pi^2 a^2 - 128\pi^2 a^2 + 2187\sqrt{7}\,a^2 + 7533 a^2 + 486 a b^4 + 128\pi^2 a b^4 - 972\sqrt{7}\,a b^3 + 4374 a b^3 - 256\sqrt{7}\,\pi^2 a b^3 - 384\pi^2 a b^3 - 6561\sqrt{7}\,a b^2 - 5589 a b^2 + 576\sqrt{7}\,\pi^2 a b^2 + 3904\pi^2 a b^2 + 8991\sqrt{7}\,a b + 31833 a b - 3008\sqrt{7}\,\pi^2 a b - 2752\pi^2 a b + 704\sqrt{7}\,\pi^2 a + 3200\pi^2 a - 8262\sqrt{7}\,a - 4617 a - 972 b^5 - 256\pi^2 b^5 + 2430\sqrt{7}\,b^4 + 1215 b^4 + 640\sqrt{7}\,\pi^2 b^4 + 320\pi^2 b^4 - 2430\sqrt{7}\,b^3 - 12636 b^3 - 640\sqrt{7}\,\pi^2 b^3 - 3328\pi^2 b^3 + 1944\sqrt{7}\,b^2 + 5346 b^2 + 512\sqrt{7}\,\pi^2 b^2 + 4288\pi^2 b^2 + 3159\sqrt{7}\,b + 15066 b - 2048\sqrt{7}\,\pi^2 b + 1088\pi^2 b - 96\sqrt{7}\,\pi^2 + 2240\pi^2 - 5832\sqrt{7} - 6804\Big) \Big/ \Big(6\big(64\pi^2 a^2 + 243 a^2 + 64\pi^2 a - 1215 a + 243 b^2 + 64\pi^2 b^2 - 243\sqrt{7}\,b - 64\sqrt{7}\,\pi^2 b + 320\pi^2 - 972\big)\Big)$$

True

True

True

```
params = tri && b > 1.08 && a > 1 / 2 /. b → Sqrt[7] / 2 - b /. a → a + 1 / 2;
(* transformed parameter space *)
RegionPlot[
 {poly2 < 0, params, 0 < a < 1 / 20 && 0 < b < 1 / 4, 5 / 100 < a < 14 / 100 && 5 / 100 < b < 1 / 4,
  14 / 100 < a < 20 / 100 && 17 / 100 < b < 1 / 4}, {a, -0.1, 0.25}, {b, -0.05, 0.3}]
```



Hence a mixed inequality is true, so must be one of the inequalities.

Proof for nearly equilateral triangles (a <= 1/2).

```
RegionPlot[{tri, 0 < a < 1 / 2 && Sqrt[3] / 2 < b < Sqrt[3]}, {a, 0, 1}, {b, 0, Sqrt[3]}]
```




```
neareq /. Or → List /. Less → Subtract
```

$$\left\{-\frac{384\,\pi^2\,b^2}{64\,\pi^2\,(a^2+b^2+3)+243\,((a-6)\,a+b^2-3)}+\left(a^2+3\right)(1-\gamma)+2\,a\,b\,\delta+b^2\,\gamma,\right.$$
$$-\frac{216\,\pi^2\,b^2}{64\,\pi^2\,(a^2+b^2+3)+243\,((a-6)\,a+b^2-3)}+\left(a^2+1\right)(1-\gamma)+2\,a\,b\,\delta+b^2\,\gamma,$$
$$\left.-\frac{192\,\pi^2\,b^2}{64\,\pi^2\,(a^2+b^2+3)+243\,((a-6)\,a+b^2-3)}+2\left(a-\frac{1}{2}\right)b\,\delta+\left(\left(a-\frac{1}{2}\right)^2+\frac{3}{4}\right)(1-\gamma)+b^2\,\gamma\right\}$$

Here we have 3 reference triangles, and no easy way of handling all cases together. In order to show that at least one inequality is true, we must show that the same is true on the boundary of 0 <= γ <= 1 and -1/2 <= δ <= 1/2. Since three lines (in γ and δ) giving the three inequalities may for a trinagle, we also need to check a point inside this triangle. The easiest way to get a point inside is to find vertices and average them.

- ### Case $\gamma = 1$:

Take the first inequality and put δ = 1/2 (worst case, positive coefficient).

```
neareq /. Or → List /. Less → Subtract;
%[[1]] /. γ → 1
rational = % /. δ → 1 / 2
RegionPlot[{rational < 0, tri && a ≤ 1 / 2 && b ≥ Sqrt[3] / 2}, {a, 0, 1}, {b, 0, Sqrt[3]}]
```

$$-\frac{384\,\pi^2\,b^2}{64\,\pi^2\,(a^2+b^2+3)+243\,((a-6)\,a+b^2-3)}+2\,a\,b\,\delta+b^2$$

$$-\frac{384\,\pi^2\,b^2}{64\,\pi^2\,(a^2+b^2+3)+243\,((a-6)\,a+b^2-3)}+a\,b+b^2$$



Note that denominator is just the upper bound, so it is positive. Hence we can take a numerator of the common fraction. We can also add any expression that is positive on the parameter space (red area).

```
poly = Numerator[Together[rational]] + 2000 (4 - (a + 1) ^2 - b^2) (b - Sqrt[3] / 2)
RegionPlot[{poly < 0, tri && a ≤ 1 / 2 && b ≥ Sqrt[3] / 2}, {a, 0, 1}, {b, 0, 2}]
```

$$243\,a^3\,b + 64\,\pi^2\,a^3\,b + 243\,a^2\,b^2 + 64\,\pi^2\,a^2\,b^2 - 1458\,a^2\,b + 243\,a\,b^3 + 64\,\pi^2\,a\,b^3 - 1458\,a\,b^2 + 2000\left(b - \frac{\sqrt{3}}{2}\right)\left(-(a+1)^2 - b^2 + 4\right) - 729\,a\,b + 192\,\pi^2\,a\,b + 64\,\pi^2\,b^4 + 243\,b^4 - 192\,\pi^2\,b^2 - 729\,b^2$$



```
pol = pol2 = poly /. b → Sqrt[3] - b; (* equilateral gives 0 for this polynomial,
so it is a good starting pooint for the algorithm *)
PolyNeg[pol, {b, a}, {2 / 3, 1 / 2}]
pol = pol /. b → b + 2 / 3;
PolyNeg[pol, {b, a}, {1 / 3, 1 / 2}]
{tri && a < 1 / 2 && b > Sqrt[3] / 2} /. b → Sqrt[3] - b (* transformed parameters *)
RegionPlot[{pol < 0, %, 0 < a < 1 / 2 && 0 < b < 2 / 3, 0 < a < 1 / 2 && 2 / 3 < b < 1},
 {a, -0.1, 1 / 2 + 0.1}, {b, -0.1, Sqrt[3] / 2 + 0.1}]
```

True

True

$$\left\{(a+1)^2+\left(\sqrt{3}-b\right)^2\leq 4\bigwedge a>0\bigwedge 2a^2+2\left(\sqrt{3}-b\right)^2>2\bigwedge a<\frac{1}{2}\bigwedge \sqrt{3}-b>\frac{\sqrt{3}}{2}\right\}$$



- **Case δ = 1/2 :**

```
neareq /. δ → 1 / 2 /. Or → List /. Less → Subtract
lin = %[[3]] // Collect[#, γ, FullSimplify] &
D[lin, γ]
```

$$\left\{-\frac{384\pi^2 b^2}{64\pi^2(a^2+b^2+3)+243((a-6)a+b^2-3)}+\left(a^2+3\right)(1-\gamma)+ab+b^2\gamma,\right.$$
$$-\frac{216\pi^2 b^2}{64\pi^2(a^2+b^2+3)+243((a-6)a+b^2-3)}+\left(a^2+1\right)(1-\gamma)+ab+b^2\gamma,$$
$$\left.-\frac{192\pi^2 b^2}{64\pi^2(a^2+b^2+3)+243((a-6)a+b^2-3)}+\left(a-\frac{1}{2}\right)b+\left(\left(a-\frac{1}{2}\right)^2+\frac{3}{4}\right)(1-\gamma)+b^2\gamma\right\}$$

$$\gamma\left(-a^2+a+b^2-1\right)-\frac{192\pi^2 b^2}{64\pi^2(a^2+b^2+3)+243((a-6)a+b^2-3)}+a^2+a(b-1)-\frac{b}{2}+1$$

$$-a^2+a+b^2-1$$

Hence coefficient for γ is positive. Put γ =2/3.

```
lin /. γ → 2 / 3
RegionPlot[{% < 0, tri && a < 1 / 2}, {a, 0, 1}, {b, 0, Sqrt[3]}]
```

$$\frac{2}{3}\left(-a^2+a+b^2-1\right)-\frac{192\,\pi^2\,b^2}{64\,\pi^2\,(a^2+b^2+3)+243\,((a-6)\,a+b^2-3)}+a^2+a\,(b-1)-\frac{b}{2}+1$$



Inequality appears to hold for γ <= 2/3. Note that denominator is positive (upper bound for eigenvalue) hance we can take just the numerator of common fraction. We can also add anything that is positive on our parameter space and negative elsewhere.

```
Together[lin /. γ → 2 / 3]
p = Numerator[%] + 10^4 (4 - (a + 1)^2 - b^2) (b - Sqrt[3] / 2)
RegionPlot[{p < 0, tri && a < 1 / 2}, {a, -0.1, 1 / 2 + 0.1}, {b, Sqrt[3] / 2 - 0.1, Sqrt[3] + 0.1}]
```

$$\left(128 \pi^2 a^4 + 486 a^4 + 1458 a^3 b + 384 \pi^2 a^3 b - 128 \pi^2 a^3 - 3402 a^3 + 1458 a^2 b^2 + 384 \pi^2 a^2 b^2 - 9477 a^2 b - 192 \pi^2 a^2 b + 512 \pi^2 a^2 + 1944 a^2 + 1458 a b^3 + 384 \pi^2 a b^3 - 6318 a b^2 - 128 \pi^2 a b^2 + 1152 \pi^2 a b - 384 \pi^2 a - 1458 a + 972 b^4 + 256 \pi^2 b^4 - 729 b^3 - 192 \pi^2 b^3 - 2430 b^2 - 256 \pi^2 b^2 + 2187 b - 576 \pi^2 b + 384 \pi^2 - 1458\right) \Big/ \left(6 \left(64 \pi^2 a^2 + 243 a^2 - 1458 a + 243 b^2 + 64 \pi^2 b^2 + 192 \pi^2 - 729\right)\right)$$

$$128 \pi^2 a^4 + 486 a^4 + 1458 a^3 b + 384 \pi^2 a^3 b - 128 \pi^2 a^3 - 3402 a^3 + 1458 a^2 b^2 + 384 \pi^2 a^2 b^2 - 9477 a^2 b - 192 \pi^2 a^2 b + 512 \pi^2 a^2 + 1944 a^2 + 1458 a b^3 + 384 \pi^2 a b^3 - 6318 a b^2 - 128 \pi^2 a b^2 + 10000 \left(b - \frac{\sqrt{3}}{2}\right) \left(-(a+1)^2 - b^2 + 4\right) + 1152 \pi^2 a b - 384 \pi^2 a - 1458 a + 972 b^4 + 256 \pi^2 b^4 - 729 b^3 - 192 \pi^2 b^3 - 2430 b^2 - 256 \pi^2 b^2 + 2187 b - 576 \pi^2 b + 384 \pi^2 - 1458$$



Again inequality seems to be true. Note that for eqilateral traingle we have equality here.

```
poly = poly2 = p /. b → Sqrt[3] - b;
PolyNeg[poly, {b, a}, {14 / 100, 25 / 100}]
poly = poly /. b → b + 14 / 100;
PolyNeg[poly, {b, a}, {31 / 100, 1 / 2}]
poly = poly /. b → b + 31 / 100;
PolyNeg[poly, {b, a}, {45 / 100, 1 / 2}]
{tri && a < 1 / 2} /. b → Sqrt[3] - b;
RegionPlot[{poly2 < 0, %, 0 < b < 14 / 100 && 0 < a < 25 / 100, 14 / 100 < b < 45 / 100 && 0 < a < 1 / 2,
  45 / 100 < b < 90 / 100 && 0 < a < 1 / 2}, {a, -0.1, 1 / 2 + 0.1}, {b, -0.1, Sqrt[3] / 2 + 0.1}]
```

True

True

True



For the case $\gamma > 2/3$ we can take the first inequality.

```
neareq /. δ → 1 / 2 /. Or → List /. Less → Subtract
lin = %[[1]] // Collect[#, γ, FullSimplify] &
D[lin, γ]
```

$$\left\{-\frac{384\,\pi^2\,b^2}{64\,\pi^2\,(a^2+b^2+3)+243\,((a-6)\,a+b^2-3)}+\left(a^2+3\right)(1-\gamma)+a\,b+b^2\,\gamma,\right.$$
$$-\frac{216\,\pi^2\,b^2}{64\,\pi^2\,(a^2+b^2+3)+243\,((a-6)\,a+b^2-3)}+\left(a^2+1\right)(1-\gamma)+a\,b+b^2\,\gamma,$$
$$\left.-\frac{192\,\pi^2\,b^2}{64\,\pi^2\,(a^2+b^2+3)+243\,((a-6)\,a+b^2-3)}+\left(a-\frac{1}{2}\right)b+\left(\left(a-\frac{1}{2}\right)^2+\frac{3}{4}\right)(1-\gamma)+b^2\,\gamma\right\}$$

$$\gamma\left(-a^2+b^2-3\right)-\frac{384\,\pi^2\,b^2}{64\,\pi^2\,(a^2+b^2+3)+243\,((a-6)\,a+b^2-3)}+a^2+a\,b+3$$

$$-a^2+b^2-3$$

Hence derivative is clearly negative. Put $\gamma = 2/3$.

```
lin /. γ → 2 / 3
RegionPlot[{% < 0, tri && a < 1 / 2}, {a, 0, 1}, {b, 0, Sqrt[3]}]
```

$$\frac{2}{3}\left(-a^2+b^2-3\right)-\frac{384\,\pi^2\,b^2}{64\,\pi^2\,(a^2+b^2+3)+243\,((a-6)\,a+b^2-3)}+a^2+a\,b+3$$



Inequality appears to hold for γ >= 2/3. Denominator is positive (upper bound for eigenvalue) hance we can take just the numerator of common fraction. We can also add anything that is positive on our parameter space and negative elsewhere.

```
Together[lin /. γ → 2 / 3]
p = Numerator[%] + 7 * 10^3 (4 - (a + 1)^2 - b^2) (b - 1)^2
RegionPlot[{p < 0, tri && a < 1 / 2}, {a, -0.1, 1 / 2 + 0.1}, {b, Sqrt[3] / 2 - 0.1, Sqrt[3] + 0.1}]
```

$$\left(64\pi^2 a^4 + 243 a^4 + 729 a^3 b + 192\pi^2 a^3 b - 1458 a^3 + 729 a^2 b^2 + 192\pi^2 a^2 b^2 - 4374 a^2 b + 384\pi^2 a^2 + 729 a b^3 + 192\pi^2 a b^3 - 2916 a b^2 - 2187 a b + 576\pi^2 a b - 4374 a + 486 b^4 + 128\pi^2 b^4 - 729 b^2 - 576\pi^2 b^2 + 576\pi^2 - 2187\right) \Big/ \left(3\left(64\pi^2 a^2 + 243 a^2 - 1458 a + 243 b^2 + 64\pi^2 b^2 + 192\pi^2 - 729\right)\right)$$

$$64\pi^2 a^4 + 243 a^4 + 729 a^3 b + 192\pi^2 a^3 b - 1458 a^3 + 729 a^2 b^2 + 192\pi^2 a^2 b^2 - 4374 a^2 b + 384\pi^2 a^2 + 729 a b^3 + 192\pi^2 a b^3 - 2916 a b^2 + 7000 (b-1)^2 \left(-(a+1)^2 - b^2 + 4\right) - 2187 a b + 576\pi^2 a b - 4374 a + 486 b^4 + 128\pi^2 b^4 - 729 b^2 - 576\pi^2 b^2 + 576\pi^2 - 2187$$



```
poly = poly2 = p /. b → Sqrt[3] - b;
PolyNeg[poly, {b, a}, {23 / 100, 33 / 100}]
poly = poly /. b → b + 23 / 100;
PolyNeg[poly, {b, a}, {43 / 100, 1 / 2}]
poly = poly /. b → b + 43 / 100;
PolyNeg[poly, {b, a}, {15 / 100, 1 / 2}]
poly = poly /. b → b + 15 / 100 /. a → a + 1 / 3;
PolyNeg[poly, {b, a}, {8 / 100, 1 / 6}]
{tri && a < 1 / 2} /. b → Sqrt[3] - b;
RegionPlot[{poly2 < 0, %, 0 < b < 23 / 100 && 0 < a < 33 / 100, 23 / 100 < b < 66 / 100 && 0 < a < 1 / 2,
   66 / 100 < b < 81 / 100 && 0 < a < 1 / 2, 81 / 100 < b < 89 / 100 && 1 / 3 < a < 1 / 2},
  {a, -0.1, 1 / 2 + 0.1}, {b, -0.1, Sqrt[3] / 2 + 0.1}]
```

True

True

True

True



- **Case δ = -1/2 :**

Take a positive linear combination of the first 2 inequalities. Note that b+a-1 is positive.

```
neareq /. δ → -1 / 2 /. Or → List /. Less → Subtract;
l1 = %[[2]]
l2 = %%[[1]]
lin = (b + a - 1) l2 + 8 / 7 (Sqrt[3] - b) l1 // Collect[#, γ, FullSimplify] &
```

$$-\frac{216\,\pi^2\,b^2}{64\,\pi^2\,(a^2+b^2+3)+243\,((a-6)\,a+b^2-3)}+\left(a^2+1\right)(1-\gamma)-a\,b+b^2\,\gamma$$

$$-\frac{384\,\pi^2\,b^2}{64\,\pi^2\,(a^2+b^2+3)+243\,((a-6)\,a+b^2-3)}+\left(a^2+3\right)(1-\gamma)-a\,b+b^2\,\gamma$$

$$\gamma\left(-\frac{8}{7}\left(a^2+1\right)\left(\sqrt{3}-b\right)-\left(a^2+3\right)(a+b-1)+b^2\,(a+b-1)+\frac{8}{7}\left(\sqrt{3}-b\right)b^2\right)+\frac{1}{7}\left(7\,a^3+a\left(b\left(-\frac{2688\,\pi^2\,b}{64\,\pi^2\,(a^2+b^2+3)+243\,((a-6)\,a+b^2-3)}+b-8\sqrt{3}+7\right)+21\right)-\frac{192\,\pi^2\,b^2\left(5\,b+9\sqrt{3}-14\right)}{64\,\pi^2\,(a^2+b^2+3)+243\,((a-6)\,a+b^2-3)}+a^2\left(-8\,b+8\sqrt{3}-7\right)+13\,b+8\sqrt{3}-21\right)$$

```
D[lin, γ] // Collect[#, a, FullSimplify] &
% < 0 /. a → 0 // Reduce // FullSimplify
%% < 0 /. a → 1 - b // Reduce // FullSimplify
```

$$-a^3+\frac{1}{7}\,a^2\left(b-8\sqrt{3}+7\right)+a\left(b^2-3\right)-\frac{1}{7}\,(b-1)\left(b\left(b-8\sqrt{3}+8\right)-8\sqrt{3}+21\right)$$

$$1<b<\sqrt{3}\ \bigvee\ b+8>7\sqrt{3}$$

$$b<1\ \bigvee\ b>\sqrt{3}$$

Note that all coefficients for powers of a are negative (knowing that b<=Sqrt[3]). We want to show that the coefficient of γ is less than 0, so plug a=0. This gives 1<b<Sqrt[3]. For a=1-b we get b<1.
Hence coefficient of γ is negative. Put γ=0.

```
Together[lin /. γ → 0]
p = Numerator[%] Denominator[%]
Flatten[{% < 0, tri && a < 1 / 2}];
RegionPlot[%, {a, 0, 1}, {b, 0, Sqrt[3]}]
```

$$\Big(448\,\pi^2 a^5 + 1701\,a^5 - 1944\,a^4 b - 512\,\pi^2 a^4 b + 512\sqrt{3}\,\pi^2 a^4 - 448\,\pi^2 a^4 + 1944\sqrt{3}\,a^4 - 11\,907\,a^4 + 1944\,a^3 b^2 + 512\,\pi^2 a^3 b^2 - 1944\sqrt{3}\,a^3 b + 13\,365\,a^3 b - 512\sqrt{3}\,\pi^2 a^3 b + 448\,\pi^2 a^3 b + 2688\,\pi^2 a^3 - 11\,664\sqrt{3}\,a^3 + 10\,206\,a^3 - 1944\,a^2 b^3 - 512\,\pi^2 a^2 b^3 + 1944\sqrt{3}\,a^2 b^2 - 3159\,a^2 b^2 + 512\sqrt{3}\,\pi^2 a^2 b^2 - 448\,\pi^2 a^2 b^2 + 11\,664\sqrt{3}\,a^2 b - 1215\,a^2 b - 704\,\pi^2 a^2 b + 2048\sqrt{3}\,\pi^2 a^2 - 2688\,\pi^2 a^2 - 3888\sqrt{3}\,a^2 - 30\,618\,a^2 + 243\,a\,b^4 + 64\,\pi^2 a\,b^4 - 1944\sqrt{3}\,a\,b^3 + 1701\,a\,b^3 - 512\sqrt{3}\,\pi^2 a\,b^3 + 448\,\pi^2 a\,b^3 + 4374\,a\,b^2 - 1152\,\pi^2 a\,b^2 + 5832\sqrt{3}\,a\,b - 24\,057\,a\,b - 1536\sqrt{3}\,\pi^2 a\,b + 1344\,\pi^2 a\,b + 4032\,\pi^2 a - 11\,664\sqrt{3}\,a + 15\,309\,a + 3159\,b^3 - 128\,\pi^2 b^3 + 1944\sqrt{3}\,b^2 - 5103\,b^2 - 1216\sqrt{3}\,\pi^2 b^2 + 1344\,\pi^2 b^2 - 9477\,b + 2496\,\pi^2 b + 1536\sqrt{3}\,\pi^2 - 4032\,\pi^2 - 5832\sqrt{3} + 15\,309\Big)\Big/ \Big(7\left(64\,\pi^2 a^2 + 243\,a^2 - 1458\,a + 243\,b^2 + 64\,\pi^2 b^2 + 192\,\pi^2 - 729\right)\Big)$$

$$7\left(64\,\pi^2 a^2 + 243\,a^2 - 1458\,a + 243\,b^2 + 64\,\pi^2 b^2 + 192\,\pi^2 - 729\right) \Big(448\,\pi^2 a^5 + 1701\,a^5 - 1944\,a^4 b - 512\,\pi^2 a^4 b + 512\sqrt{3}\,\pi^2 a^4 - 448\,\pi^2 a^4 + 1944\sqrt{3}\,a^4 - 11\,907\,a^4 + 1944\,a^3 b^2 + 512\,\pi^2 a^3 b^2 - 1944\sqrt{3}\,a^3 b + 13\,365\,a^3 b - 512\sqrt{3}\,\pi^2 a^3 b + 448\,\pi^2 a^3 b + 2688\,\pi^2 a^3 - 11\,664\sqrt{3}\,a^3 + 10\,206\,a^3 - 1944\,a^2 b^3 - 512\,\pi^2 a^2 b^3 + 1944\sqrt{3}\,a^2 b^2 - 3159\,a^2 b^2 + 512\sqrt{3}\,\pi^2 a^2 b^2 - 448\,\pi^2 a^2 b^2 + 11\,664\sqrt{3}\,a^2 b - 1215\,a^2 b - 704\,\pi^2 a^2 b + 2048\sqrt{3}\,\pi^2 a^2 - 2688\,\pi^2 a^2 - 3888\sqrt{3}\,a^2 - 30\,618\,a^2 + 243\,a\,b^4 + 64\,\pi^2 a\,b^4 - 1944\sqrt{3}\,a\,b^3 + 1701\,a\,b^3 - 512\sqrt{3}\,\pi^2 a\,b^3 + 448\,\pi^2 a\,b^3 + 4374\,a\,b^2 - 1152\,\pi^2 a\,b^2 + 5832\sqrt{3}\,a\,b - 24\,057\,a\,b - 1536\sqrt{3}\,\pi^2 a\,b + 1344\,\pi^2 a\,b + 4032\,\pi^2 a - 11\,664\sqrt{3}\,a + 15\,309\,a + 3159\,b^3 - 128\,\pi^2 b^3 + 1944\sqrt{3}\,b^2 - 5103\,b^2 - 1216\sqrt{3}\,\pi^2 b^2 + 1344\,\pi^2 b^2 - 9477\,b + 2496\,\pi^2 b + 1536\sqrt{3}\,\pi^2 - 4032\,\pi^2 - 5832\sqrt{3} + 15\,309\Big)$$



Hence inequality should hold.

```
poly = p /. b → Sqrt[3] - b;
PolyNeg[poly, {b, a}, {1 / 5, 12 / 100}]
poly = poly /. b → b + 1 / 5;
PolyNeg[poly, {b, a}, {35 / 100, 12 / 100}]

poly = p /. b → Sqrt[3] / 2 + b /. a → 1 / 2 - a;
PolyNeg[poly, {b, a}, {Sqrt[3] / 2, 38 / 100}]

poly = p /. b → Sqrt[3] / 2 + b /. a → 12 / 100 - a;
PolyNeg[poly, {b, a}, {32 / 100, 12 / 100}]

Flatten[{p < 0, tri, Sqrt[3] - 20 / 100 < b < Sqrt[3] && 0 < a < 12 / 100,
   Sqrt[3] - 55 / 100 < b < Sqrt[3] - 20 / 100 && 0 < a < 12 / 100,
   Sqrt[3] / 2 < b < Sqrt[3] / 2 + 32 / 100 && 0 < a < 12 / 100,
   Sqrt[3] / 2 < b < Sqrt[3] && 12 / 100 < a < 1 / 2}];
RegionPlot[%, {a, 0, 1}, {b, 0, Sqrt[3]}]
```

True

True

True

True



- **Case $\gamma = 0$:**

Here we can take a linear combination of inequalities 2 and, in such a way that δ gets cancelled.

```
neareq /. γ -> 0 /. r → List /. Less → Subtract;
l1 = %[[2]]
l2 = %%[[3]]
lin = a l2 + (1 / 2 - a) l1 // Collect[#, γ, FullSimplify] &
RegionPlot[{% < 0, tri && a < 1 / 2}, {a, -0.1, 1 / 2 + 0.1}, {b, Sqrt[3] / 2 - 0.1, Sqrt[3] + 0.1}]
```

$$-\frac{216\,\pi^2\,b^2}{64\,\pi^2\,(a^2+b^2+3)+243\,((a-6)\,a+b^2-3)}+a^2+2\,a\,b\,\delta+1$$

$$-\frac{192\,\pi^2\,b^2}{64\,\pi^2\,(a^2+b^2+3)+243\,((a-6)\,a+b^2-3)}+2\left(a-\frac{1}{2}\right)b\,\delta+\left(a-\frac{1}{2}\right)^2+\frac{3}{4}$$

$$\frac{-243\,(a^2-1)\,((a-6)\,a+b^2-3)-8\,\pi^2\,((8\,a^2-6\,a+19)\,b^2+8\,(a^4+2\,a^2-3))}{128\,\pi^2\,(a^2+b^2+3)+486\,((a-6)\,a+b^2-3)}$$



Inequality seems true. Again we take just the numerator.

```
p = Numerator[lin];
poly = p /. b → 97 / 100 + b;
PolyNeg[poly, {b, a}, {8 / 10, 1 / 2}]
poly = p /. b → Sqrt[3] / 2 + b /. a → a + 1 / 5;
PolyNeg[poly, {b, a}, {1 / 5, 3 / 10}]
RegionPlot[{p < 0, tri && a < 1 / 2, 97 / 100 < b < 177 / 100 && 0 < a < 1 / 2,
   Sqrt[3] / 2 < b < Sqrt[3] / 2 + 1 / 5 && 1 / 5 < a < 1 / 2},
  {a, -0.1, 1 / 2 + 0.1}, {b, -0.1 + Sqrt[3] / 2, Sqrt[3] + 0.1}]
```

True

True



■ Interior point :

Here we take all pairs of inequlities as equations of lines in (γ, δ) and set them equal 0. We get vertices of the triangle formed by the lines. Average of the vertices gives a center of mass which is inside of the triangle. One of the inequalities must be true at this point if the point is in the parameter space.

```
lines = neareq /. Less → Subtract /. Or → List
{γ, δ} /. (lines[[1]] == lines[[2]] == 0 // Solve[#, {γ, δ}][[1]] & // FullSimplify)
{γ, δ} /. (lines[[2]] == lines[[3]] == 0 // Solve[#, {γ, δ}][[1]] & // FullSimplify)
{γ, δ} /. (lines[[1]] == lines[[3]] == 0 // Solve[#, {γ, δ}][[1]] & // FullSimplify)
(% + %% + %%%) / 3 // FullSimplify
neareq /. γ → %[[1]] /. δ → %[[2]] /. Or → List // FullSimplify
p = %[[3, 1]]
RegionPlot[{p > 0, tri}, {a, 0, 1}, {b, 0, Sqrt[3]}]
```

$$\Big\{-\frac{384\,\pi^2\,b^2}{64\,\pi^2\,(a^2+b^2+3)+243\,((a-6)\,a+b^2-3)}+\left(a^2+3\right)(1-\gamma)+2\,a\,b\,\delta+b^2\,\gamma,$$

$$-\frac{216\,\pi^2\,b^2}{64\,\pi^2\,(a^2+b^2+3)+243\,((a-6)\,a+b^2-3)}+\left(a^2+1\right)(1-\gamma)+2\,a\,b\,\delta+b^2\,\gamma,$$

$$-\frac{192\,\pi^2\,b^2}{64\,\pi^2\,(a^2+b^2+3)+243\,((a-6)\,a+b^2-3)}+2\left(a-\frac{1}{2}\right)b\,\delta+\left(\left(a-\frac{1}{2}\right)^2+\frac{3}{4}\right)(1-\gamma)+b^2\,\gamma\Big\}$$

$$\left\{1-\frac{84\,\pi^2\,b^2}{64\,\pi^2\,(a^2+b^2+3)+243\,((a-6)\,a+b^2-3)},\ \frac{4\,\pi^2\,b\,(5\,(b^2-3)-37\,a^2)-243\,b\,((a-6)\,a+b^2-3)}{128\,\pi^2\,a\,(a^2+b^2+3)+486\,a\,((a-6)\,a+b^2-3)}\right\}$$

$$\left\{\frac{243\,(a^2-1)\,((a-6)\,a+b^2-3)+8\,\pi^2\,((8\,a^2-6\,a+19)\,b^2+8\,(a^4+2\,a^2-3))}{(a^2+b^2-1)\,(64\,\pi^2\,(a^2+b^2+3)+243\,((a-6)\,a+b^2-3))},\right.$$
$$\left.\frac{-243\,a\,b\,((a-6)\,a+b^2-3)-8\,\pi^2\,b\,(a\,(a\,(8\,a+3)+8\,b^2-3)-3\,b^2+3)}{(a^2+b^2-1)\,(64\,\pi^2\,(a^2+b^2+3)+243\,((a-6)\,a+b^2-3))}\right\}$$

$$\left\{\frac{64\,\pi^2\,(a^4+4\,a^3+((a-2)\,a+3)\,b^2+12\,a-9)+243\,(a\,(a+4)-3)\,((a-6)\,a+b^2-3)}{(a\,(a+4)+b^2-3)\,(64\,\pi^2\,(a^2+b^2+3)+243\,((a-6)\,a+b^2-3))},\right.$$
$$\left.\frac{-243\,(a+2)\,b\,((a-6)\,a+b^2-3)-64\,\pi^2\,b\,(a\,(a\,(a+5)+b^2-3)-b^2+3)}{(a\,(a+4)+b^2-3)\,(64\,\pi^2\,(a^2+b^2+3)+243\,((a-6)\,a+b^2-3))}\right\}$$

$$\Big\{\Big(4\,\pi^2\,\big(48\,(a-1)\,(a+1)\,(a^2+3)\,(a\,(a+4)-3)+2\,(a\,(19\,a-32)+77)\,b^4+(a\,(a\,(a\,(91\,a+128)+58)+624)-561)\,b^2-5\,b^6\big)+$$
$$243\,\big((a-6)\,a+b^2-3\big)\big(4\,(a\,(a+2)-2)\,b^2+3\,(a-1)\,(a+1)\,(a\,(a+4)-3)+b^4\big)\Big)\Big/$$
$$\Big(3\,\big(a^2+b^2-1\big)\big(a\,(a+4)+b^2-3\big)\big(64\,\pi^2\,\big(a^2+b^2+3\big)+243\,\big((a-6)\,a+b^2-3\big)\big)\Big),$$
$$-\Big(243\,b\,\big((a-6)\,a+b^2-3\big)\big(2\,a\,(3\,a+4)\,b^2+a\,(a\,(a\,(5\,a+16)-12)-8)+b^4-4\,b^2+3\big)+$$
$$4\,\pi^2\,b\,\big((a\,(91\,a-64)+35)\,b^4+a\,(a\,(a\,(197\,a+384)-382)+256)\,b^2+$$
$$a\,(a\,(a\,(a\,(a\,(101\,a+448)-321)-224)+231)-192)-5\,b^6-75\,b^2+45\big)\Big)\Big/$$
$$\Big(6\,a\,\big(a^2+b^2-1\big)\big(a\,(a+4)+b^2-3\big)\big(64\,\pi^2\,\big(a^2+b^2+3\big)+243\,\big((a-6)\,a+b^2-3\big)\big)\Big)\Big\}$$

$$\Big\{b^2\,\big(a^2+b^2-1\big)\big(64\,\pi^2\,\big(a^2+b^2+3\big)+243\,\big((a-6)\,a+b^2-3\big)\big)\big(4\,\pi^2\,\big(a\,(5\,a-12)+5\,\big(b^2-3\big)\big)-243\,\big((a-6)\,a+b^2-3\big)\big)>0,$$
$$b^2\,\big(a\,(a+4)+b^2-3\big)\big(64\,\pi^2\,\big(a^2+b^2+3\big)+243\,\big((a-6)\,a+b^2-3\big)\big)\big(4\,\pi^2\,\big(a\,(5\,a-12)+5\,\big(b^2-3\big)\big)-243\,\big((a-6)\,a+b^2-3\big)\big)<0,$$
$$a\,b^2\,\big(64\,\pi^2\,\big(a^2+b^2+3\big)+243\,\big((a-6)\,a+b^2-3\big)\big)\big(4\,\pi^2\,\big(a\,(5\,a-12)+5\,\big(b^2-3\big)\big)-243\,\big((a-6)\,a+b^2-3\big)\big)>0\Big\}$$

$$a\,b^2\,\big(64\,\pi^2\,\big(a^2+b^2+3\big)+243\,\big((a-6)\,a+b^2-3\big)\big)\big(4\,\pi^2\,\big(a\,(5\,a-12)+5\,\big(b^2-3\big)\big)-243\,\big((a-6)\,a+b^2-3\big)\big)$$



Hence the third inequyality should be true at the center of mass (even if it is not inside of the parameter space). Note that this polynomial is a product of ab^2, upper bound for eigenvalue and the last factor which should be positive.

```
poly = -p[[4]] /. b → Sqrt[3] - b // Collect[#, {b, a}, FullSimplify] &
PolyNeg[poly, {b, a}, {Sqrt[3] / 2, 1 / 2}]
```

$$\left(243 - 20\,\pi^2\right) a^2 + 6\left(8\,\pi^2 - 243\right) a + \left(243 - 20\,\pi^2\right) b^2 + 2\,\sqrt{3}\,\left(20\,\pi^2 - 243\right) b$$

True

Spectral gap for an acute triangle

In[1]:=
```
AppendTo[$Path, ToFileName[{$HomeDirectory, "Dropbox", "mathematica"}]];
<< TrigInt`
```

Here we can numerically optimize a test function, rationalize coefficient and find symbolic upper bound. We can also find optimal lower bound based on at least 3 reference triangles.

Tables of eigenfunctions

Sorted eigenfunctions for known cases.

```
? Equilateral
? Square
```

Eigenfunctions of Neumann and Dirichlet Laplacian
on the equilateral triangle with vertices (0,0), (1,0), (1/2,Sqrt[3]/2):
Equilateral[Dirichlet,Symmetric] [m,n] – 1<=m<=n
Equilateral[Dirichlet,Antisymmetric] [m,n] – 1<=m<n
Equilateral[Neumann,Symmetric] [m,n] – 0<=m<=n
Equilateral[Neumann,Antisymmetric] [m,n] – 0<=m<n
Equilateral[Eigenvalue] [m,n]
Eigenfunctions of the right triangle with vertices (0,0), (1,0), (0,Sqrt[3]):
Equilateral[Dirichlet,Half] [m,n] – 1<=m<n
Equilateral[Neumann,Half] [m,n] – 0<=m<=n
Equilateral[Eigenvalue,Half] [m,n]
Vertices:
Equilateral[]
Equilateral[Half]

Eigenfunctions of Neumann and Dirichlet Laplacian on the square with vertices (0,0), (1,0), (1,1), (0,1):
Square[Dirichlet] [m,n] – m>=1,n>=1
Square[Neumann] [m,n] – m>=0,n>=0
Square[Eigenvalue] [m,n]
Eigenfunctions of the right isosceles triangle with vertices (0,0), (1,0), (0,1):
Square[Dirichlet,Half] [m,n] – 1<=m<n
Square[Neumann,Half] [m,n] – 0<=m<=n
Vertices:
Square[Half] ≫

Equilateral

In[3]:=
```
num = 13;
Table[{Equilateral[Eigenvalue][a, b], a, b}, {a, 0, num}, {b, 0, num}];
max = %[[1, -1, 1]];
%% // Flatten[#, 1] & // Sort[#, #1[[1]] <= #2[[1]] &] & // Select[#, #[[1]] ≤ max &] &;
Rest /@ % // Rest;
eqeigfun = If[#1 ≤ #2, Equilateral[Neumann, Symmetric][#1, #2],
       Equilateral[Neumann, Antisymmetric][#1, #2]] & @@ # & /@ %;
Length[
 eqeigfun]
```

Out[9]= 115

■ 30 - 60 - 90

In[10]:=
```
num = 17;
Table[{Equilateral[Eigenvalue, Half][a, b], a, b}, {a, 0, num}, {b, 0, num}];
max = %[[1, -1, 1]];
%% // Flatten[#, 1] & // Sort[#, #1[[1]] <= #2[[1]] &] & //
  Select[#, #[[2]] ≤ #[[3]] && #[[1]] ≤ max &] &;
Rest /@ % // Rest;
heigfun = Equilateral[Neumann, Half] @@ # & /@ %;
Length[heigfun]
```

Out[16]= 100

■ Right isosceles

In[17]:=
```
num = 15;
Table[{Square[Eigenvalue][a, b], a, b}, {a, 0, num}, {b, 0, num}];
max = %[[1, -1, 1]]
%% // Flatten[#, 1] & // Sort[#, #1[[1]] <= #2[[1]] &] & //
  Select[#, #[[2]] ≤ #[[3]] && #[[1]] ≤ max &] &;
Rest /@ % // Rest;
rieigfun = Square[Neumann, Half] @@ # & /@ %;
Length[rieigfun]
```

Out[19]= $225\pi^2$

Out[23]= 101

■ Optimal upper bound.

Take the first ? eigenfunctions for the 3 known cases and transplant them. I should also implement rotated transplantations.

In[673]:=
```
T2[a_, b_] = {{-1, 0}, {1, 0}, {a, b}};
T2[a_, b_, cond_] = {{-1, 0}, {1, 0}, {a, b}, cond};
ri = h = eq = 5;
Take[eqeigfun, eq];
Transplant[#, RotateLeft[T2[a, b]], Equilateral[]] & /@ %;
eqc = Table[ceq[i], {i, eq}];
eqtest = eqc.%%;

Take[heigfun, h];
Transplant[#, Reverse[T2[a, b]], Equilateral[Half]] & /@ %;
hc = Table[ch[i], {i, h}];
htest = hc.%%;

Take[rieigfun, ri];
Transplant[#, Reverse[T2[a, b]], Square[Half]] & /@ %;
ric = Table[cri[i], {i, ri}];
ritest = ric.%%;

Clear[c]
test = eqtest + htest + ritest;
TrigInt[test, Limits[T2[a, b, b > 0]]] (* this integrates to 0 *)
(*Integrate[test,Limits[T2[a,b,b>0]],Assumptions→b>0] (* this is very slow *)*)

Timing[bound = Rayleigh[test, T2[a, b, b > 0]]][[1]]
```

Out[690]= 0

Out[691]= 36.7235

In[378]:=
```
Clear[trifun]
triangle = (#[[2]] ≤ #[[1]] ≤ #[[3]] & /@ {Limits[T2[a, b, b > 0]]}) /. List → And
trifun[c_, d_] := Function[{xx, yy}, Evaluate[triangle /. a → c /. b → d /. x → xx /. y → yy]]
```

Out[379]= $0 \leq y \leq b \bigwedge \frac{(a+1)\,y}{b} - 1 \leq x \leq \frac{(a-1)\,y}{b} + 1$

In[622]:=
```
optimal[c_?NumericQ, d_?NumericQ] :=
 NMinimize[{bound, Max[Abs[ceq[1]], Abs[ch[1]], Abs[cri[1]]] == 1 && ceq[1] + ceq[2] == 1} /.
     a → c /. b → d, Flatten[{eqc, hc, ric}]][[2]] // Rationalize[#, 0.000001] &
optupper[c_, d_] := {test, bound} /. optimal[c, d] /. a → c /. b → d;
optupper[2 / 1000, 17 / 10];
ContourPlot[%[[1]], {x, -1, 1}, {y, 0, 2},
 RegionFunction → trifun[aa, bb], Contours → 40, AspectRatio → Automatic]
4 %%[[2]] // N
```

Out[625]=



Out[626]= 17.6212

- **Optimal lower bound.**

In[64]:=
```
ineq1 = (a^2 + 3) (1 - γ) + 2 a b δ + b^2 γ < 4 π^2 / 3 / u;
ineq2 = (a^2 + 1) (1 - γ) + 2 a b δ + b^2 γ < 3 π^2 / 4 / u;
ineq3 = ((a - 1 / 2) ^2 + 3 / 4) (1 - γ) + 2 (a - 1 / 2) b δ + b^2 γ < 2 π^2 / 3 / u;
ineq4 = ((a - 1) ^2 + 4 / 3) (1 - γ) + 2 (a - 1) b δ + b^2 γ < 8 π^2 / 9 / u;
```

```
In[275]:= ineqs = ineq1 || ineq2 || ineq3 || ineq4 /. u → π^2 / u /. Less → Equal //
            Collect[#, {γ, δ}, FullSimplify] &
          lessu = u /. (% // Solve[#, u] &)
          optimal2[c_?NumericQ, d_?NumericQ] :=
           Maximize[Evaluate[{Min[lessu], 0 ≤ γ ≤ 1 && -1 / 2 ≤ δ ≤ 1 / 2} /. a → c /. b → d], {γ, δ}][[1]]
          optlower[c_, d_] := lower = 2 u /. u → π^2 / optimal2[c, d]
          optlower[2 / 1000, 17 / 10] // N
```

Out[275]= $\gamma\left(-a^2+b^2-3\right)+a^2+2\,a\,b\,\delta+3=\frac{4\,u}{3}\bigvee\gamma\left(-a^2+b^2-1\right)+a^2+2\,a\,b\,\delta+1=\frac{3\,u}{4}\bigvee$
$\gamma\left(-a^2+a+b^2-1\right)+(2\,a-1)\,b\,\delta+(a-1)\,a+1=\frac{2\,u}{3}\bigvee\gamma\left(-(a-2)\,a+b^2-\frac{7}{3}\right)+2\,(a-1)\,b\,\delta+(a-1)^2+\frac{4}{3}=\frac{8\,u}{9}$

Out[276]= $\left\{-\frac{3}{4}\left(a^2\gamma-a^2-2\,a\,b\,\delta-b^2\gamma+3\,\gamma-3\right),-\frac{4}{3}\left(a^2\gamma-a^2-2\,a\,b\,\delta-b^2\gamma+\gamma-1\right),\right.$
$\left.-\frac{3}{2}\left(a^2\gamma-a^2-2\,a\,b\,\delta-a\,\gamma+a-b^2\gamma+b\,\delta+\gamma-1\right),-\frac{3}{8}\left(3\,a^2\gamma-3\,a^2-6\,a\,b\,\delta-6\,a\,\gamma+6\,a-3\,b^2\gamma+6\,b\,\delta+7\,\gamma-7\right)\right\}$

Out[279]= 8.8855

■ Lower bound for the gap for triangle (-1,0), (1,0) and (a,b)

Gives exact value for equilateral, right isosceles and 30 - 60 - 90.

```
In[692]:= gap[c_, d_] := optlower[c, d] - 2 optupper[c, d][[2]]
          4 gap[0, Sqrt[3]] // N
          4 gap[0, Sqrt[3] * 9999 / 10 000] // N
          4 gap[2 / 1000, 17 / 10] // N
          4 gap[1 / 4, Sqrt[2]] // N
          optupper[1 / 4, Sqrt[2]][[2]] 4 // N
```

Out[693]= 0.

Out[694]= 0.000257758

Out[695]= 0.299767

Out[696]= 5.29651

Out[697]= 17.4199